:: ===========================================
:: 此脚本用于激活 Windows 10 Enterprise 批量授权版
:: ===========================================

@echo off
echo 开始获取管理员权限...

%1 %2
mshta vbscript:createobject("shell.application").shellexecute("%~s0","goto :runas","goto :notadmin","runas",1)(window.close)&&exit

:notadmin
echo 获取管理员权限失败，请右键点击此文件，选择“以管理员身份运行”。
pause
EXIT

:runas
:: KMS 服务器地址
SET KMSAddress=X.X.X.X
:: KMS 服务器端口，默认为 1688，通常无需修改。
SET KMSPort=1688
:: Windows 10 Enterprise 的 GVLK 密钥
SET GVLK=NPPR9-FWDCX-D2C8J-H872K-2YT43
echo 已获取管理员权限，将开始为产品进行授权...
echo=
echo 开始卸载现有产品密钥...
cscript //Nologo %windir%\system32\slmgr.vbs -upk
echo 开始安装新的产品密钥...
cscript //Nologo %windir%\system32\slmgr.vbs -ipk %GVLK%
:: 指定 KMS（密钥管理服务） 服务器的 IP 地址或域名
echo 开始设置 KMS 服务器...
cscript //Nologo %windir%\system32\slmgr.vbs -skms %KMSAddress%
:: 指定 KMS（密钥管理服务）服务器的 TCP 通信端口
cscript //Nologo %windir%\system32\slmgr.vbs -sprt %KMSPort%
echo 开始激活 Windows 10 企业版...
cscript //Nologo %windir%\system32\slmgr.vbs -ato
:: 显示当前许可证的有效日期
cscript //Nologo %windir%\system32\slmgr.vbs -xpr
pause>null
exit